MAPLE SUPPLEMENT, REVIEW 3, CALCULUS 1

PROBLEM 1

We start with defining our function

```
> f:=x->(x+x^2)*exp(x);
```

$$f := x \to \left(x + x^2\right) e^x \quad (1)$$

Next we compute the first derivative

```
> diff(f(x),x);
```

$$(1 + 2x)\, e^x + \left(x + x^2\right) e^x \quad (2)$$

We simplify the expression

```
> simplify(%);
```

$$e^x \left(1 + 3x + x^2\right) \quad (3)$$

And now we can find the stationary points

```
> solve(x^2+3*x+1);
```

$$-\frac{3}{2} + \frac{1}{2}\sqrt{5},\ -\frac{3}{2} - \frac{1}{2}\sqrt{5} \quad (4)$$

We can also find decimal approximations for these points

```
> fsolve(x^2+3*x+1);
```

$$-2.618033989,\ -0.3819660113 \quad (5)$$

To see the character of the stationary points with MAPLE the easiest thing is to graph the polynomial $1 + 3x + x^2$.

```
> plot(x^2+3*x+1, x=-4..1);
```



This graph tells us where the first derivative is positive (and the original function is increasing) or negatiove (the original function is decreasing). It also tells us that the point $-\frac{3}{2} - \frac{1}{2}\sqrt{5}$ is a relative maximum and the point $-\frac{3}{2} + \frac{1}{2}\sqrt{5}$ is a relative minimum.

PROBLEM 2

We compute the second derivative

```
> diff(f(x),x$2);
```

$$2\,e^x + 2\,(1+2\,x)\,e^x + (x+x^2)\,e^x \quad (6)$$

```
> simplify(%);
```

$$e^x\,(4+5\,x+x^2) \quad (7)$$

The inflection points are

```
> solve(x^2+5*x+4);
```

$$-1,\ -4 \quad (8)$$

```
> plot(x^2+5*x+4, x=-5..1);
```



The graph above shows that the original function is concave up on the intervals (-∞,-4) and (-1,∞), and concave down on the interval (-4,-1).

PROBLEM 3

We can now graph the function on an interval including all "points of interest".

```
> plot(f(x),x=-5..0.5);
```



PROBLEM 4

To find the MacLaurin polynomial of degree 3 of function *f* we perform the command

```
> taylor(f(x),x=0,4);
```

$$x + 2x^2 + \frac{3}{2}x^3 + O(x^4) \tag{9}$$

Pay attention that the number at the end of the command, in our case 4, should be one more than the degree of the polynomial.
Now we get rid of the last term

```
> convert(%, polynom);
```

$$x + 2x^2 + \frac{3}{2}x^3 \tag{10}$$

and evaluate the value of this polynomial at 0.1

```
> evalf(1.5*0.1^3+2*0.1^2+0.1);
```

$$0.1215 \tag{11}$$

Finally to estimate the error of approximation we can use the command

```
> maximize(abs(diff(f(x),x$4))/4!*0.1^4, x=0..0.1);
```

$$0.00007786850094 \tag{12}$$

We can also check our result in the following way

```
> evalf(f(0.1)-0.1215);
```

$$0.0000688010 \tag{13}$$

which shows that the real error is indeed a bit smaller than our estimate.

PROBLEM 5

On the test you should provide the complete solution of a similar problem but you can check your result as follows

```
> limit((1+cos(x))^tan(x),x=Pi/2);
```

$$e \tag{14}$$

PROBLEM 6

We introduce

```
> g:=x->3*sec(x)+4*csc(x);
```

$$g := x \to 3\sec(x) + 4\csc(x) \tag{15}$$

To find the stationary point of this function we can use the command

```
> solve([diff(g(x),x)=0, x>0, x< Pi/2], x);
```

$$\left\{x = \arctan\left(\frac{1}{3}\,6^{2/3}\right)\right\} \tag{16}$$

The value of the function at the stationary point is

```
> g(arctan((1/3)*6^(2/3)));
```

$$\sqrt{9 + 6\,6^{1/3}} + \frac{2}{3}\sqrt{9 + 6\,6^{1/3}}\;6^{1/3} \tag{17}$$

Numerical approximations

```
> evalf(arctan((1/3)*6^(2/3)));
```

$$0.8332718597 \tag{18}$$

```
> evalf(g(arctan((1/3)*6^(2/3))));
```

$$9.865662558 \tag{19}$$

PROBLEM 7

We introduce the function *P(x)*

```
> P:=x->x^3+x^2-1;
```

$$P := x \to x^3 + x^2 - 1 \tag{20}$$

and the function *G(x)=x - P(x)/P'(x)*

```
> G:=x->x-P(x)/D(P)(x);
```

$$G := x \to x - \frac{P(x)}{D(P)(x)} \tag{21}$$

The initial approximation

```
> 0.5;
```

$$0.5 \tag{22}$$

The consecutive approximations

```
> evalf(G(%));
```

$$0.8571428571 \quad (23)$$

```
> evalf(G(%));
```

$$0.7641369047 \quad (24)$$

```
> evalf(G(%));
```

$$0.7549634825 \quad (25)$$

```
> evalf(G(%));
```

$$0.7548776738 \quad (26)$$

```
> evalf(G(%));
```

$$0.7548776663 \quad (27)$$

```
> evalf(G(%));
```

$$0.7548776663 \quad (28)$$

PROBLEM 8

We will use implicit differentiation. We assume that the height of the cone is a function of its radius. Than the volume will be expressed as

```
> V:=r -> Pi*r^2*h(r);
```

$$V := r \to \pi r^2 h(r) \quad (29)$$

and therefore the derivative of V by r is

```
> diff(V(r),r);
```

$$2 \pi r h(r) + \pi r^2 \left(\frac{d}{dr} h(r) \right) \quad (30)$$

Because V is a fixed constant this derivative is identically 0 and we get the following expression for the derivative of h by r

```
> solve(diff(V(r),r),diff(h(r),r));
```

$$-\frac{2 h(r)}{r} \quad (31)$$

Next we introduce the surface area of the cone as a function of r

```
> S:=r -> Pi*r*sqrt(r^2+h(r)^2);
```

$$S := r \to \pi r \sqrt{r^2 + h(r)^2} \quad (32)$$

The derivative of S is

```
> diff(S(r),r);
```

$$\pi \sqrt{r^2 + h(r)^2} + \frac{1}{2} \frac{\pi r \left(2 r + 2 h(r) \left(\frac{d}{dr} h(r) \right) \right)}{\sqrt{r^2 + h(r)^2}} \quad (33)$$

In this expression we substitute the derivative $\frac{d}{dr} h(r)$ by the expression $-\frac{2 h(r)}{r}$.

```
> subs(diff(h(r), r)=-2*h(r)/r, diff(S(r),r));
```

$$\pi \sqrt{r^2 + h(r)^2} + \frac{1}{2} \frac{\pi r \left(2 r - \frac{4 h(r)^2}{r} \right)}{\sqrt{r^2 + h(r)^2}} \quad (34)$$

At a stationary point the last expression must be 0; thus to find the relation between h and r at a stationary point we perform the command

```
> solve(%,h(r));
```

$$\sqrt{2}\, r,\ -\sqrt{2}\, r \tag{35}$$

Sign minus obviously does not make sense in our problem and therefore we have the relation

```
> h:=r-> sqrt(2)*r;
```

$$h := r \rightarrow \sqrt{2}\, r \tag{36}$$

Finally we can find the optimal values of r , h, and S

```
> r:=(V/Pi/sqrt(2))^(1/3);
```

$$r := \frac{1}{2}\, 2^{2/3} \left(\frac{\sqrt{2}\, V}{\pi}\right)^{1/3} \tag{37}$$

```
> h:=r*sqrt(2);
```

$$h := 2^{1/6} \left(\frac{\sqrt{2}\, V}{\pi}\right)^{1/3} \tag{38}$$

```
> S:= Pi*r*sqrt(r^2+h^2);
```

$$S := \frac{1}{4}\, \pi\, 2^{2/3} \left(\frac{\sqrt{2}\, V}{\pi}\right)^{1/3} \sqrt{6} \sqrt{2^{1/3} \left(\frac{\sqrt{2}\, V}{\pi}\right)^{2/3}} \tag{39}$$

THE END